Fermilab
ES&H Section

E.P. NOTE 7

CALCULATION OF MAXIMAL TRITIUM CONCENTRATIONS IN SOIL NEAR AP0

E. Marshall and K. Vaziri

February 1997
(Revised May 1999)

Author: [signature] Date: 5/17/99
E. Marshall

Author: [signature] Date: 5/17/99
K. Vaziri

Approved: [signature] Date: 5/18/99
D. Cossairt
Associate Head, Radiation Protection

Approved: [signature] Date: 5/18/99
R. Walton
Associate Head, Environment, Safety & Assurance

Approved: [signature] Date: 5-18-99
Bill Griffing
Head, ES&H Section

Distribution via e-mail
EP and RP Group Staff
Authors

E.P. NOTE 7

CALCULATION OF MAXIMAL TRITIUM CONCENTRATIONS IN SOIL NEAR AP0

E. Marshall and K. Vaziri

February 1997
(Revised May 1999)

Introduction

A program of study has been underway to investigate the radionuclide migration in soil around the AP0 target and several other areas to provide additional data for the groundwater concentration model adopted in December 1994 [Ref. 1]. The area around AP0 was selected for study for a number of reasons, primarily because of the stability of the beam parameters, the relatively large number of protons targeted during its operation, and the existence of the recorded operational history of.the antiproton source.

This note is a summary of the modeling and calculations performed by the authors and the conclusions drawn from the information gathered.

Geometry Summary

The geometry used in the CASIM simulations was compiled from shielding configuration drawings, drawings of the target assembly, lithium lens and pulsed magnet, conversations with engineers and scientists assigned responsibility for AP0, and actual measurements, in addition to the numerous papers published on the antiproton source. (Refs. 2 - 20) In the actual coding phase of this project, simplifications were made to the geometry so that the code could run efficiently. Experience with CASIM has shown that this level of geometry simplification causes only minor differences in the calculations.

For future reference, a copy of the geometry used in the CASIM simulations is included in the Appendix, as is a copy of one of the data files and magnetic field file. Figure 1 depicts the basic geometry of the vault and beam absorber.

Simulations

CASIM simulations were run to obtain sufficient statistics to generate star density curves. For these calculations, the star densities were calculated using the CASIM defaults for radial and longitudinal binning. Figure 2 is a reproduction of a typical star density contour plot.

The results of the CASIM simulations are provided in cylindrical coordinates. However, for the purposes of this study, it was necessary to have the results in rectangular coordinates. Several CASIM simulations were performed to compare the star density at a particular location in the shield to that determined from the contour plots of azimuthally averaged star density so the activity gradient in the soil could be predicted. To do this, the geometry code was modified to allow special bins to be set up adjacent to the structural wall and also immediately within the soil shield. With the total number of stars within the bin and the volume of the bin, the star density at a particular location could be determined and compared to the star density obtained using the contour plots. Within a factor of 5, the azimuthally averaged star densities indicated on the star density contour plots could be used to estimate the rectangular star densities.

Discussion

With the star density calculated by the CASIM runs, it is possible, knowing the number of incident protons and their distribution in time, to calculate the radionuclide concentrations. The radionuclide of greatest concern is tritium (half-life of 12.3 years). The total number of protons targeted was obtained from the operating records of the antiproton source. The proton intensities are summarized in the Table 1 below. [Ref. 21; Ref. 22]

Table 1: Total Number of Protons on Target by Calendar Year

Year	Number of Protons ($x10^{18}$)
1986	0.4
1987	3.4
1988	5.8
1989	4.7
1990	2.58
1991	2.22
1992	5.99
1993	6.56
1994	17.90
1995	14.49
1996	3.58
1997	3.55
TOTAL	**71.2**

Since tritium concentrations in the soil are proportional to the number of stars produced, it was important to see whether or not sufficient stars were being produced to result in detectable tritium concentrations in an area where a soil boring might be possible. Compensating for the radioactive decay of tritium over the time profile, the integrated effective number of protons targeted, or the number of protons that would need to be delivered at once, is equal to 5.3 x 10^{19}. Assuming a tritium concentration of 1 pCi/g, which is approximately the limit of detection [Ref. 23], a soil density of 2.25 g/cm^3, a conversion factor of 0.075 ^{3}H atoms/star [Ref. 24], the star density would have to be on the order of 1 x 10^{-11} stars/cm^3/incident proton. In comparing this star density with those obtained from the CASIM calculations performed, this star density is seen at distances of less than approximately 260 cm (8.5 ft) from the beam center perpendicular to the target, at distances of less than 380 cm (12.6 ft) from the beam center perpendicular to the pulse magnet and at distances less than 340 cm (11.2 ft) from the beam center in the vicinity of the beam absorber. For reference, the shielding and structural concrete around the beam go out to a minimum radial distance of 240 cm or 7.9 ft from the beam center. Unfortunately for purposes of testing soil migration, the AP0 Service Building completely covers the volume of soil where a detectable tritium concentration could be expected. However, the presence of the building diverts water from this area and renders migration to the groundwater even less likely.

The star density plots were also used to determine what tritium concentrations could be expected immediately outside the concrete structural walls and also at the outer boundary of the gravel barrier which lies between the walls and the soil. Several distances along the beam axis were selected for these calculations. Table 2 presents the results of this series of calculations. The star density included in Table 2 is the average of the star densities from star density plots and the error is equal to their standard deviation. Figure 1 illustrates the basic geometry of the vault for reference.

Table 2: Expected Tritium Concentrations at Selected Locations Outside the AP0 Vault

Z (cm)	Location	R (cm)	Star Density (stars/cm^3/incident particle)	Tritium Conc. (pCi/cm^3)
250	Across from target, outside wall	244	(1.32 ± 0.17)E-11	1.1 ± 0.1
	Across from target, outside gravel	305	(4.05 ± 0.53)E-12	0.35 ± 0.05
400	Across from magnet, outside wall	244	(1.22 ± 0.17)E-10	10.5 ± 1.5
	Across from magnet, outside gravel	305	(2.66 ± 0.27)E-11	2.3 ± 0.2
680	Across from beam absorber, outside wall	350	(7.79 ± 3.58)E-12	0.67 ± 0.31
	Across from beam absorber, outside gravel	380	(4.67 ± 0.79)E-12	0.40 ± 0.07



Figure 1: Basic Geometry of AP0 Vault and Beam Absorber (not to scale)

Discussion and Conclusions

This simulation used extremely conservative assumptions. In addition, the area is protected from rainwater by a roof. Class I groundwater is present only at depths of approximately 40 feet of very low conductivity till, whereas the predictions were made for a depth of about 13 feet. Over a period of several years, water samples collected from the sumps positioned in this area have been analyzed for tritium levels. These measurements, presented in Table 3, agree well with the calculated predictions and are consistently below 100 pCi/ml, given the standard deviations associated with the star densities. It is recognized that there are additional errors associated with the predictions, such as the uncertainty associated with the number of incident protons. However, it is believed that these are relatively small in comparison.

Table 3: Tritium Measurements at AP0

LOCATION	SAMPLE ID	^{3}H CONCENTRATION (pCi/ml)	ERROR (pCi/ml)	% ERROR
AP0 Prevault	910531	4.87	0.45	9.2%
	910907	26.35	0.88	3.3%
	911118	3.4	0.39	11.5%
	920221	23.311	0.686	2.9%
	920611	5.807	0.566	9.7%
	920827	5.887	0.604	10.3%
	921102	1.361	0.328	24.1%
	930309	0	0.415	
	930611	2.341	0.377	16.1%
	930812	2.340	0.380	16.2%
	931119	6.195	0.633	10.2%
	940309	2.483	0.800	32.2%
	940309	2.9	0.2	6.9%
	940720	18.9	1.1	5.8%
	940830	10.3	1.0	9.7%
	941214	2.15	0.98	45.6%
	950220	2.70	0.60	22.2%
	950613	4.68	0.69	14.7%
	950828	70.09	1.80	2.6%
	951212	6.52	0.83	12.7%
	960307	3.3	0.7	21.2%
	960611	4.43	0.73	16.5%
	961209	77.2	1.4	1.8%
	930708	0	0.375	
AP0 Transport Line	940406	0.24	0.80	333.3%
	940406	0	1.0	
	950216	0	0.56	
	960321	1.2	0.7	58.3%

All the calculations predict tritium concentrations much less that the regulatory limit for Class I drinking water (20 pCi/ml), with the exception of the value across from the magnet and immediately outside the vault wall (Z=400 cm and R=244 cm). This all suggests it would be unlikely that much information could be gleaned about the vertical migration of the tritium by performing borings in the vicinity of the AP0 target and beam absorber. These results also imply the probability of high levels of tritium migrating to the aquifer is vanishingly small. The next step would be to bore immediately downstream of the AP0 Service building as close to beam center as possible to verify these calculations.

References

1. Cossairt, J. Donald. Use of a Concentration-Based Model for Calculating the Radioactivation of Soil and Groundwater at Fermilab. E.P. Note #8. December 1994.

2. Yurista, P. Soil Activation Antiproton Target Station. $\bar{p}$ Note #295.

3. Canada, C. V. An Updated Review of Soil Radioactivation Calculations for the Antiproton Target Area. Fermilab-TM-1023. December 1980.

4. Cossairt, J. D. and P. Yurista. Shielding Calculations for the Antiproton Target Area. Fermilab-TM-1136. September 1982.

5. Hojvat, C., et. al. The Fermilab Tevatron I Project Target Station for Antiproton Production. Fermilab-TM-1174. March 1983.

6. Freeman, W. S. Energy Deposition in the $\bar{p}$ Target Hall Proton Beam Dump. $\bar{p}$ Note #357. January 1984.

7. Reilly, R. E. Procedure for Changing the AP2 Beam Stop. March 8, 1985.

8. Memo to M. Gormley from P. Yurista. Request for Autopsy of Antiproton Production Target. January 7, 1988.

9. Bhat, C. Energy Deposition and Radiation Shielding in the Pbar Source at Fermilab. Fermilab-conf-92/83. March 1992.

10. Malensek, A. J. Neutrino Beams Using the Main Injector. Fermilab-TM-1951. November 1995.

11. Antiproton Source Shielding Layout. Drawing Number: 8055-ME-119393. August 10, 1984.

12. Antiproton Source Shielding Steel Placement Layout Volume 1. Drawing Number: 8055-ME-119382. August 9, 1984.

13. Antiproton Source Shielding Steel Placement Layout Volume 2. Drawing Number: 8055-ME-119383. August 9, 1984.

14. Antiproton Source Shielding Steel Placement Layout Volume 3. Drawing Number: 8055-ME-119384. August 9, 1984.

15. Anitproton Source Shielding Steel Placement Layout Volume 4. Drawing Number: 8055-ME-119385. August 9, 1984.

16. Anitproton Source Shielding Steel Placement Layout Volume 5. Drawing Number: 8055-ME-119386. August 9, 1984.

17. Antiproton Source Shield Steel Placement Elevation, Detailing and Optional Shimming. Drawing Number: 8055-ME-119398. August 21, 1984.

18. Antiproton Source Lithium Lens Transformer General Assembly. Drawing Number: 8000-ME-322037. April 8, 1993.

19. Antiproton Source Assembly Target Module Mark II. Drawing Number: 8000-ME-254137 Rev. C. April 1994.

20. Antiproton Source Target Pulse Magnet #3 Main Assembly. Drawing Number: 8055.000-ME-322170. May 14, 1994.

21. Private communication. Mike Church to Don Cossairt. MR Beam on target '86-'89. June 29, 1996.

22. Memos, ES&H Section Head to Distribution. Tabulation of Off-Site Radiological Parameters. 1990-1996.

23. Private communication. Vernon Cupps to Elaine Marshall. Detection Limit for Tritium in Soil. August 1996.

24. Gollon, P. J. Soil Activation Calculations for the Anti-Proton Target Area. Fermilab Report TM-816. 1978.



R-LABELS REFER TO SMALLER VALUES OF CORRESPONDING BINS
LEGEND : NUMERICAL SYMBOLS REFER TO NEGATIVE POWER OF 10, E.G., 5 REFERS TO THE 10**-5 CONTOUR
OTHER POWERS OF 10(SYMBOLS) :-10(A),-11(B),-12(C),-13(D),-14(E),-15(F),-16(G),-17(H),-18(I),-19(J)
1(Z),2(Y),3(X),4(W),5(V),6(U),7(T),8(S),9(R),10(Q)

Figure 2: Equal Star Density Contour Plot. The horizontal axis corresponds to the distance along the beam axis or Z. Z=0 was arbitrarily placed at the entrance to the target vault. The vertical axis corresponds to the radial distance from the beam center (R=0).

APPENDIX

FORTRAN CODES

APPENDIX

Geometry Coding

```
SUBROUTINE CASIMGEOM
C
C     USER SUBROUTINE DESCRIBING PROBLEM GEOMETRY
C
C     GIVEN ( X,Y,Z ) IN CM, ENTRY USRGEOM RETURNS
C           MATERIAL INDEX N (GE.-1  AND LE.9)
C           MAGNETIC REGION INDEX M (GE.0)
C
C     CONVENTIONS NM=-1 VACUUM WITH MAGNETIC FIELD PRESENT
C                 0 VACUUM  NO  MAGNETIC FIELD PRESENT
C                 1-9 MATERIALS
C                  99 Outside of defined geometry
C
C     MAGNETIC FIELD INDEX M MUST CORRESPOND TO INDEX IN
C     SUBROUTINE USRFIELD WHERE B VECTOR IS DEFINED AS A FUNCTION OF
C     LOCATION
C
      IMPLICIT NONE
      SAVE
C
C========================== Include HIBI.CIN ============================
C HIBI
C
      REAL ZLIM, RLIM
C
      COMMON/HIBI/ ZLIM, RLIM
C============================ End Include ===============================
C
      INTEGER N, M, KT
C
C L = LENS
C M = MAGNET
C S = SHIELD
C V = TARGET VAULT
C D = DUMP VAULT
C T = TARGET
C SPPRT=SUPPORT
C C = CONCRETE
C GRAV=GRAVEL
C
      REAL X, Y, Z
      REAL XM, YM, ZM
      REAL BX, BY, BZ
      REAL R
      REAL Z1, Z2, TSTART, Z4, Z5, TEND, LSTART, Z8, Z9, LEND, MSTART
      REAL MEND,Z13
      REAL Z14, VAULTEND, Z16, Z17, Z18, DSTART, Z20, DEND, ALDEND
      REAL MAGCTR
      REAL XAPP1, YAPP1, XAPP2, YAPP2TOP, YAPP2BOT, XAPP3
      REAL VSRT, VSLT, VSBOT, VSTOP, SPPRTBOT, SPPRTTOP
      REAL VSLEDGE
      REAL VTOP, VBOT, VRT, VLT
      REAL VCTOP, VCBOT, VCRT, VCLT
      REAL VGRAVBOT, VGRAVRT, VGRAVLT
      REAL TOPSOIL
      REAL XT, YT
      REAL TSPPRTLT, TSPPRTY
      REAL LENSRAD, LFRAMERAD, LSUPPORTX, LSUPPORTY
      REAL MAGXAPP, MAGYAPP, MAGWIDTH, MAGTOP, MAGBOT
      REAL DSRT, DSLT, DSBOT, DSTOP, LEDGEBOT
      REAL DSTORTOP
```

Geometry Coding

```
      REAL DVTOP, DVRT, DVLT1, DVLT2, DVBOT
      REAL DCTOP1, DCTOP2, DCBOT, DCLT1, DCLT2, DCLT3, DCRT
      REAL DGRAVBOT, DGRAVRT, DGRAVLT1, DGRAVLT2
      REAL DUMPRAD,ANNULUS
      REAL BSYAPPTOP, BSYAPPBOT
      REAL AX, AY
      REAL OFFSET
      REAL BSTOPLT, BSTOPRT
C
C-----------------------------------------------------------------
C Hard-wire the limits to your geometry here.  This section gets
C called once per Job.  The 'SAVE' above insures that values set
C are retained after the subroutine is exited.
C
C Define various Z and geometry limits.
      Z1=7.62
      Z2=41.91
      TSTART=244.44
      Z4=247.61
      Z5=252.69
      TEND=255.86
      LSTART=270.32
      Z8=274.88
      Z9=289.88
      LEND=294.45
      MSTART=340.19
      MEND=452.74
      Z13=459.66
      Z14=466.01
      VAULTEND=500.30
      Z16=505.38
      Z17=562.61
      Z18=602.62
      DSTART=623.57
      Z20=741.68
      DEND=743.57
      ALDEND=843.57
C
      MAGCTR=396.47
C
      ZLIM=1127.76
C
      RLIM=1200.0
C
C Define geometry limits for target vault.
C
C Voids in shielding to allow beam to pass through.
C
      XAPP1 = 15.25
      YAPP1 = 15.25
C
      XAPP2 = 39.35
      YAPP2TOP = 45.7
      YAPP2BOT = -52.1
C
C Exterior dimensions of target vault shield.
C
      VSRT=137.2
      VSLT=-137.2
      VSBOT=-198.1
      VSTOP=228.6
C
      SPPRTBOT=66.1
```

Geometry Coding

```
      SPPRTTOP=218.5
C
C VSHIELDLEDGE=VSHIELDTOP-45.7
      VSLEDGE=182.9
C
C Exterior dimensions of vault area.
C
      VTOP=411.5
C VAULTBOT=VSHIELDBOT
      VBOT=-198.1
      VRT=182.9
      VLT=-167.6
C
C Exterior dimensions of concrete surrounding vault.
C
      VCTOP=600.8
C VCBOT=VAULTBOT-91.4
      VCBOT=-289.5
C VCRT=VAULTRT+76.2
      VCRT=259.1
C VCLT=VAULTLT-76.2
      VCLT=-243.8
C
C Exterior dimensions of gravel placed outside of concrete walls.
C
C VGRAVBOT=VCBOT-61.0
      VGRAVBOT=-350.5
C VGRAVRT=VCRT+152.4
      VGRAVRT=411.5
C VGRAVLT=VCLT-152.4
      VGRAVLT=-396.2
C
C Level of ground relative to beam height.
      TOPSOIL=457.2
C
C Target dimensions.
      XT=5.72
      YT=7.6
C
C Dimensions of support apparatus for target.
C
C TSUPPORTLT=-(43.50-XT)
      TSPPRTLT=-37.78
C TSUPPORTY=13.97+YT
      TSPPRTY=21.57
C
C Dimensions of Li lens and approximated support.
C
      LENSRAD=1.0
      LFRAMERAD=22.9
C LENSSUPPORTY=LENSFRAMERAD+6.99
      LSUPPORTY=29.89
      LSUPPORTX=38.4
C
C Dimensions of pulsed magnet (current).
      MAGXAPP=2.55
      MAGYAPP=1.91
C
      MAGWIDTH=16.55
      MAGTOP=22.5
      MAGBOT=-12.7
C
C Define geometry limits for dump.
```

Geometry Coding

```
C
      XAPP3 = 20.5
C
      DSRT=131
      DSLT=-144.7
      DSBOT=-198.1
      DSTOP=233.7
C
C DSTORTOP=top of concrete fill in hot storage area
C
      DSTORTOP=182.86
C
C LEDGEBOT=DSHIELDTOP-41.9
      LEDGEBOT=191.8
C
      DVTOP=274.3
      DVRT=182.9
      DVLT1=-167.6
      DVLT2=-274.3
C DVAULTBOT=DSHIELDBOT
      DVBOT=-198.1
C
      DCTOP1=541.6
      DCTOP2=556.9
C DCBOT=DVAULTBOT-91.4
      DCBOT=-289.5
C DCLT1=DVAULTLT1-152.4
      DCLT1=-320.0
C DCLT2=DVAULTLT2-76.2
      DCLT2=-350.5
      DCLT3=-167.6
C DCRT=DVAULTRT+91.4
      DCRT=274.3
C
C DGRAVBOT=DCBOTTOM-61.0
C DGRAVRT=DCRT+61.0
C DGRAVLT1=DCLT1-61.0
C DGRAVLT2=DCLT2-61.0
      DGRAVBOT=-350.5
      DGRAVRT=335.3
      DGRAVLT1=-381
      DGRAVLT2=-411.5
C
      DUMPRAD=7.62
      ANNULUS=15.24
C
      BSYAPPTOP=43.2
      BSYAPPBOT=-30.5
C
      RETURN
C
C-----------------------------------------------------------------
C
C This section gets called whenever CASIM wants to check what material
C a particle is in, given X, Y, Z
C
      ENTRY USRGEOM(X,Y,Z,N,M,KT,XM,YM,ZM)
C
C Keep these 5 lines here, always,  CASIM needs the counter kT, and needs to
C know when the counter has gone beyond its limit.
      KT=KT+1
      IF(KT.GT.10000) THEN
        N=99
```

Geometry Coding

```
        RETURN
      ENDIF
C
C Set N=0 until we know if we are in some other material.
      N=0
C
C Set M=0 until we know if we are in a field region.
      M=0
C
C Define comparison variables.
C
      AX=ABS(X)
      AY=ABS(Y)
C
      R=SQRT(X*X + Y*Y)
C
C OFFSET expands width of void around dump for secondary beam.
C OFFSET= -XAPP3 - ((Z-MAGCENTER) * TAN(0.0523))
      OFFSET=0.235 - 0.0523*Z
C
C BSTOP are the variables that position beam stop of secondary beam.
C BSTOPRT=(TAN(0.0523) * (Z-MAGCENTER)) + 4.45
C BSTOPLT=(TAN(0.0523) * (Z-MAGCENTER)) - 4.45
      BSTOPRT=0.0523*Z - 16.29
      BSTOPLT=0.0523*Z - 25.2
C
C First check the limits of the geometry and exit if outside them
C
      IF (R .GT. RLIM) THEN
         N = 99
         RETURN
      ENDIF
C
      IF (Z .LE. 0. .OR. Z .GT. ZLIM) THEN
         N = 99
       RETURN
      ENDIF
C
C List materials used.
C N=0 Air
C N=1 Steel
C N=2 Concrete
C N=3 Gravel
C N=4 Soil
C N=5 Target/Cu
C N=6 Li
C N=7 Fe/Cu
C N=8 Graphite
C N=9 Aluminum
C
C Geometry coding below.
C
C Set "Magnet" coordinates the same as Lab coordinates.
      XM=X
      YM=Y
      ZM=Z
C
      IF (Z .GE. 0 .AND. Z .LE. VAULTEND) THEN ! In target vault?
C
C Code for shielding at start of vault & before 1st support plate.
C
         IF (Z .LE. Z1) THEN
C
```

APPENDIX

Geometry Coding

```
C Check if beam opening.
C
            IF (AX .GT. XAPP1 .OR. AY .GT. YAPP1) THEN
              N=1
            ENDIF
C
C Outside shield but still in vault?
C
            IF (AX .GT. VSRT .OR. Y .GT. SPPRTBOT) THEN
              N=0
            ENDIF
         ENDIF
C
C Code for support plates.
C
         IF ((Z .GT. Z1 .AND. Z .LE. Z2) .OR. Z .GT. Z14) THEN
C
C Check if outside beam opening.
C
            IF (AX .GT. XAPP1 .OR. AY .GT. YAPP1) THEN
              N=1
            ENDIF
C
C Outside shield but still in vault?
C
            IF (AX .GT. VSRT .AND. Y .LT. SPPRTBOT) THEN
              N=0
            ENDIF
C
C Atop shielding but still in vault?
C
            IF (Y .GT. SPPRTTOP) THEN
              N=0
            ENDIF
         ENDIF
C
C Code for larger aperture opening.
C
         IF (Z .GT. Z2 .AND. Z .LE. Z13) THEN
C
C Code for target.
C
            IF (Z .GT. TSTART .AND. Z .LE. TEND) THEN
C
C Check within target.
C
               IF (AX .LE. XT .AND. AY .LE. YT) THEN
                 N=5
               ENDIF
C
C Code for target support
C
               IF (Z .GT. Z4 .AND. Z .LE. Z5) THEN
C
                  IF((X.GT.TSPPRTLT.AND.X.LT.XT).AND.
     +               (Y.GT.YT.AND.Y.LT.TSPPRTY))THEN
                     N=1
                  ENDIF
C
               ENDIF
            ENDIF
C
C Code for Li Lens
```

APPENDIX

Geometry Coding

```
C
          IF (Z .GT. LSTART .AND. Z .LE. LEND) THEN
             IF (R .LT. LENSRAD) THEN ! Code for Li Lens itself.
                IF (Z .GT. Z8 .AND. Z .LE. Z9) THEN
                   N=6
                   M=1
                   CALL USRFIELD(XM,YM,ZM,M,BX,BY,BZ)
                ENDIF
             ENDIF
C
C Code for Li Lens frame, cooling, and electronics.
C
             IF (R .GE. LENSRAD .AND. R .LT. LFRAMERAD) THEN
                N=1
             ENDIF
C
C Code for lens support.
C
             IF (Y .GT. LFRAMERAD .AND. Y .LT. LSUPPORTY) THEN
                N=1
             ENDIF
          ENDIF
C
C Code for pulsed magnet.
C
          IF (Z .GT. MSTART .AND. Z .LE. MEND) THEN
C
             IF (AX .LE. MAGXAPP .AND. AY .LE. MAGYAPP) THEN
                M=2
                CALL USRFIELD(XM,YM,ZM,M,BX,BY,BZ)
             ENDIF
C
C Within magnet aperture?
C
             IF (AX .GT. MAGXAPP .OR. AY .GT. MAGYAPP) THEN
                N=7
             ENDIF
C
C Within magnet?
C
             IF (AX .GT. MAGWIDTH .OR. Y .GT. MAGTOP .OR.
     +           Y .LT. MAGBOT) THEN
                N=0
             ENDIF
          ENDIF
C
C Check to see if in shielding.
C
          IF (AX.GE.XAPP2.OR.Y.GT.YAPP2TOP.OR.Y.LT.YAPP2BOT) THEN
             N=1
          ENDIF
C
C Check to see if outside shielding but still in vault.
C
          IF (X .LT. VSLT .OR. Y .GT. VSTOP) THEN
             N=0
          ENDIF
C
          IF (X .GT. VSRT .AND. Y .LT. VSLEDGE) THEN
             N=0
          ENDIF
C
       ENDIF ! end code for larger aperture.
```

Geometry Coding

```
C
C Code for space before second support plate.
C
        IF (Z .GT. Z13 .AND. Z .LE. Z14) THEN
C
C Check beam opening.
C
           IF (AX .GT. XAPP1 .OR. AY .GT. YAPP1) THEN
              N=1
           ENDIF
C
C Check to see if outside shielding but still in vault.
C
           IF (AX .GT. VSRT .OR. Y .GT. VSTOP) THEN
              N=0
           ENDIF
        ENDIF
C
C Code for concrete around target vault.
C
        IF (X.GT.VRT .OR. X.LT.VLT .OR. Y.GT.VTOP .OR. Y.LT.VBOT) THEN
           N=2
        ENDIF
C
C Code for gravel surrounding vault.
C
        IF (X .GT. VCRT .OR. X .LT. VCLT .OR. Y .LE. VCBOT) THEN
           N=3
        ENDIF
C
C Code for soil around vault.
C
        IF (X.GT.VGRAVRT .OR. X.LT.VGRAVLT .OR. Y.LE.VGRAVBOT) THEN
           N=4
        ENDIF
C
C Code for void outside vault.
C
        IF ((X .GT. VCRT .OR. X .LT. VCLT) .AND. Y .GT. TOPSOIL) THEN
           N=0
        ENDIF
C
C Code for void above vault.
C
        IF (X .LE. VCRT .AND. X .GE. VCLT .AND. Y .GT. VCTOP) THEN
           N=0
        ENDIF
C
     ENDIF ! End of geometry for target vault.
C
C What follows is the code for the geometry of the area in
C front of the dump.
C
     IF (Z .GT. VAULTEND .AND. Z .LE. Z18) THEN
C
        IF (Z .LT. Z16 .OR. Z .GT. Z17) THEN
C
C Beam pipe in shielding.
C
           IF (AX .GT. XAPP1 .OR. AY .GT. YAPP1) THEN
              N=1
           ENDIF
        ENDIF
```

APPENDIX

Geometry Coding

```
C
         IF (Z .GT. Z16 .AND. Z .LT. Z17) THEN ! Check for beam opening.
            IF (AX .GT. XAPP3 .OR. AY .GT. YAPP1) THEN
               N=1
            ENDIF
         ENDIF
C
C Checks to see if outside shielding but still in vault.
C
         IF (X .LT. DSLT .OR. Y .GT. DSTOP) THEN
            N=0
         ENDIF
C
         IF (X .GT. DSRT .AND. Y .LT. LEDGEBOT) THEN
            N=0
         ENDIF
C
C Check to see if within concrete walls.
C
         IF (X.GT.DVRT.OR.X.LT.DVLT1.OR.Y.GT.DVTOP.OR.Y.LT.DVBOT) THEN
            N=2
         ENDIF
C
C Check to see if within gravel around the vault.
C
         IF (X .GT. DCRT .OR. X .LT. DCLT1 .OR. Y .LT. DCBOT) THEN
            N=3
         ENDIF
C
C Check to see if within soil around vault.
C
         IF (X.GT.DGRAVRT .OR. X.LT.DGRAVLT1 .OR. Y.LT.DGRAVBOT) THEN
            N=4
         ENDIF
C
C Check to see if in void outside vault.
C
         IF ((X .GT. DCRT .OR. X .LT. DCLT1) .AND. Y .GT. TOPSOIL) THEN
            N=0
         ENDIF
C
C Check to see if in void above vault.
C
         IF (X .LE. DCRT .AND. X .GE. DCLT1 .AND. Y .GT. DCTOP1) THEN
            N=0
         ENDIF
C
      ENDIF ! End of geometry for shielding in front of beam dump.
C
C Code for target dump and beam stop follows.
C
      IF (Z .GT. Z18) THEN
C
C Code for target dump.
C
         IF (Z .LE. DSTART) THEN
            IF (AX .GT. XAPP1 .OR. AY .GT. YAPP1) THEN
               N=1 ! Assign material to steel outside beam opening.
            ENDIF
         ENDIF
C
C Dump before opening enlarges.
C
```

Geometry Coding

```
        IF (Z .GT. DSTART .AND. Z .LE. Z20) THEN
           IF (R .LT. DUMPRAD) THEN ! Check to see if in beam dump.
              N=8
           ENDIF
C
           IF (R .LT. ANNULUS .AND. R .GE. DUMPRAD) THEN
              N=9
           ENDIF
C
C Check if in shielding.
C
           IF (AX .GT. XAPP3 .OR. AY .GT. YAPP1) THEN
              N=1
           ENDIF
        ENDIF
C
C Dump after opening enlarges.
C
        IF (Z .GT. Z20 .AND. Z .LE. ALDEND) THEN
C
           IF (Z .LE. DEND) THEN
              IF (R .LT. DUMPRAD) THEN ! Check if in beam dump.
                 N=8
              ENDIF
              IF (R .LT. ANNULUS .AND. R .GE. DUMPRAD) THEN
                 N=9
              ENDIF
C
C Check if in Al section of dump.
C
           ELSE
              IF (R .LT. ANNULUS) THEN
                 N=9
              ENDIF
C
           ENDIF
C
           IF (X .GT. XAPP3 .OR. X .LT. OFFSET .OR. AY .GT. YAPP1) THEN
              N=1
           ENDIF
        ENDIF
C
C Code for beam stop.
C
        IF (Z .GT. ALDEND) THEN ! Code for beam stop void space.
           N=1
           IF (X.LT.BSTOPRT .AND. X.GT.BSTOPLT .AND. Y.GT.BSYAPPBOT
     +        .AND. Y.LT.BSYAPPTOP) THEN
              N=0
           ENDIF
        ENDIF
C
C Check to see if outside shielding but still within vault.
C
        IF (X .LT. DSLT .AND. Y .LT. DSTORTOP) THEN
           N=2
        ENDIF
C
        IF (X .LT. DSLT .AND. Y .GE. DSTORTOP) THEN
           N=0
        ENDIF
C
```

Geometry Coding

```
        IF (X .GT. DSRT .AND. Y .LT. LEDGEBOT) THEN
           N=0
        ENDIF
C
C Check to see if in concrete walls.
C
        IF(X.GT.DVRT.OR.X.LT.DVLT2.OR.Y.GT.DVTOP.OR.Y.LT.DVBOT)THEN
           N=2
        ENDIF
C
C Check to see if in gravel around vault.
C
        IF (X .GT. DCRT .OR. X .LT. DCLT2 .OR. Y .LT. DCBOT) THEN
           N=3
        ENDIF
C
C Check to see if in soil around vault.
C
        IF (X.GT.DGRAVRT .OR. X.LT.DGRAVLT2 .OR. Y.LE.DGRAVBOT) THEN
           N=4
        ENDIF
C
C Check to see if in void outside vault.
C
        IF ((X .GT. DCRT .OR. X .LT. DCLT2) .AND. Y .GT. TOPSOIL) THEN
           N=0
        ENDIF
C
C Check to see if in void between concrete shielding atop and
C concrete walls on sides of vault.
C
        IF (X .LT. DCLT3 .AND. X .GT. DVLT2 .AND. Y .GT. DVTOP) THEN
           N=0
        ENDIF
C
C Check to see if in void above vault.
C
        IF (X .LE. DCRT .AND. X .GE. DCLT2 .AND. Y .GT. DCTOP2) THEN
           N=0
        ENDIF
C
      ENDIF ! End of geometry for beam dump and beam stop region.
C
      RETURN
      END
```

Example Data File

```
! The first lines hold the list of customized files you need to run
! your CASIM geometry.  The two listed here are the MINIMUM you need, and
! should always be there.  If you gave the CASIMGEOM file a customized name,
! then you need to edit that name here in place of CASIMGEOM.FOR.
! Add more files as needed, add them as new lines below these two,
! 1 file-name per line, left-justified.
cASIM.FOR
ap0GEOM.FOR
geomview.for
usrFIELD.FOR
!!!!!!!!!!!! The CASIM Input Deck information starts below this line !!!!!
         1      120.      0.3       0.2       0.2       1.0      1000000
    0    2    1    1 293847591  1.E-18     10.0      15.0
         9
    26.0      55.8      7.87      288.       1.80      4.97      0.385
    11.4      22.9      2.40      143.0     10.90      3.69      0.171
    11.4      22.9      2.00      143.0     13.08      3.69      0.171
    10.7      21.6      2.25      135.      12.30      3.62      0.162
    29.0      63.5      8.96      317.0      1.47      5.19      0.433
     3.0       6.9      0.53       52.2    152.1       3.4       0.057
    26.4      56.9      8.02      292.0      1.74      5.00      0.392
     6.0      12.0      2.62       86.0     43.0       2.98      0.095
    13.0      27.0      2.70      159.0      8.99      3.90      0.199
!!!!!! but is provided to list the standard parameters for materials used
!!!!!! by most CASIM geometries.  Copy lines below into the Input Deck above
!!!!!! as needed.  Do NOT copy the preceeding material ID line.  The order
!!!!!! in which they are listed does not imply any assumptions within CASIM.
!!!!!! For other materials not given below, see TM-1898.
! Fe  Iron
    26.0      55.8      7.87      288.0      1.80      4.97       0.385
! Cu  Copper
    29.0      63.5      8.96      317.0      1.47      5.19       0.433
! Cu/H2O combined; for water cooled magnet coils
    29.0      63.5      8.20      317.0      1.60      5.19       0.433
! Brass:  Mixture of 70% Copper and 30% Zinc
    29.3      64.0      8.50      320.0      1.53      5.20       0.436
! Al  Aluminum
    13.0      27.0      2.70      159.0      8.99      3.90       0.199
! Be  Berylium
     4.0       9.0      1.85       64.0     35.10      3.40       0.073
! W   Tungstun
    74.0     183.9     19.30      746.0      0.35      7.39       1.140
! Pb  Lead
    82.0     207.2     11.35      821.0      0.56      7.69       1.270
! Moist Soil
    10.7      21.6      2.25      135.0     12.30      3.62       0.162
! Ordinary Concrete
    11.4      22.9      2.40      143.0     10.90      3.69       0.171
```

APPENDIX

Magnetic Field File

```
SUBROUTINE USRFIELD (XM, YM, ZM, NMAG, BX, BY, BZ)
C
C     XM,YM,ZM ARE COORDS RELATIVE TO MAGNET (LABELED NMAG)
C       ROUTINE RETURNS FIELD COMPS BX,BY,BZ (KGAUSS) REL TO LAB COORDS
C
      IMPLICIT NONE
C
      INTEGER NMAG
      REAL XM, YM, ZM, BX, BY, BZ
C
C-------------------------------------------------------

      BX=0.
      BY=0.
      BZ=0.
C
      IF(NMAG.EQ.0) RETURN
C
C
      IF (NMAG .EQ. 1) THEN
         BX=100*YM
         BY=-100*XM
      ELSE IF (NMAG .EQ. 2) THEN
         BY=-0.465
      ENDIF
C
      RETURN
      END
```